
Neste documento estará disposto um exemplo de layout de etiqueta gerado a partir 

de um script contendo comandos em SBPL (Linguagem de programação SATO) e 

toda a descrição de cada linha contida neste script. 

Caracteres de controle: 

São caracteres utilizados para realizar alguma ação: Definir configurações da 

impressora (tamanho da etiqueta, deslocamento de imagem, parâmetros de RFID, 

etc); envio de impressão (campos impressos e quantidade). 

Os caracteres de controle são: STX, ETX, ESC, SOH, EOT, ENQ, DC2, CAN, NUL. 

Estes caracteres de controle têm que ser inseridos mediante suas identificações em 

decimal ou hexadecimal segundo a tabela ASCII. A tabela abaixo descreve os 

respectivos caracteres na codificação UTF-8: 

Caractere Código Decimal Código Hexadecimal 

[STX] 2 02 

[ETX] 3 03 

[ESC] 27 1B 

[SOH] 1 01 

[EOT] 4 04 

[ENQ] 5 05 

[DC2] 18 12 

[CAN] 24 18 

[NUL] 0 00 

O manual de programação em SBPL identifica o caractere ESC por meios dos 

caracteres < e >. Isto significa que ao longo do manual de programação, por 

exemplo, o comando ESC+A é identificado por <A>. 

 
• EXEMPLO DE IMPRESSÃO VIA COMANDOS SBPL: 

 

 
Layout de teste gerado por comandos em SBPL 

 
 

Deste ponto em diante será descrito o script contendo os comandos que 

possibilitaram a geração da etiqueta acima. 

Script utilizado: 

<STX> 
<DC2>PI,SB 
<ESC>A 
<ESC>IP5190,220 
<ESC>IG0 
<ESC>PH0 
<ESC>PM2 
<ESC>PO0-27 
<ESC>A3V+00000H+0000 
<ESC>CS4 
<ESC>#F7 
<ESC>A1V00400H0799 
<ESC>Z 
<ETX> 
<STX> 
<ESC>A 
<ESC>RU,01 
<ESC>IP0e:h,epc:0123456789ABCDEF,fsw:0; 
<ESC>PS 
<ESC>%0 
<ESC>H0155<ESC>V00078<ESC>P02<ESC>RH0,SATOSANS.ttf,0,074,079,TESTE SATO 
<ESC>H0199<ESC>V00190<ESC>BG04080>I123456789012 
<ESC>H0301<ESC>V00274<ESC>P02<ESC>RDB@0,038,038,123456789012 
<ESC>H0186<ESC>V00348<ESC>P02<ESC>RH0,SATOSANS.ttf,0,033,028,EPC: 
<ESC>H0275<ESC>V00348<ESC>P02<ESC>RH0,SATOSANS.ttf,0,033,028,0123456789ABCDEF 
<ESC>H0081<ESC>V00073<ESC>FW0303V0250H0639 
<ESC>H0024<ESC>V00023<ESC>GB019004DADOS_IMAGEM 
<ESC>Q1 
<ESC>Z 
<ETX> 
<DC2>PK 

Explicação dos comandos utilizados: 
 
A ordem será a seguinte: Primeiro será mostrado o comando e na linha de baixo, 
estará o comentário iniciando com “//”, demonstrando que é um comentário. 
 
 
• Definição de configurações gerais e RFID da impressora: 

 
<STX> 
// Caractere de controle que indica início de job 
 
<DC2>PI,SB 
// Define a linguagem da impressora ->  00 : AUTO     
      SB : SBPL 
      SZ : SZPL 
      SI : SIPL 
      SD : SDPL 
      ST : STCL 
      SE : SEPL 
 
<ESC>A 
// Inicio de sequência 
 
<ESC>IP5190,220 
// Define potência de escrita e leitura RFID ( IP5aaa,bbb ) 
             -> aaa : potência de escrita ( 000 a 240 ) 
             -> bbb : potência de leitura ( 000 a 240 ) 
             
            OBS.1: 000 equivale a 0 dBm 
            OBS.2: 240 equivale a 24 dBm 
 
<ESC>IG0 
/ Define o sensor utilizado -> 0 : I-Mark 

     -> 1 : Gap 
        -> 2 : Sensor desabilitado 
 
<ESC>PH0 
// Define o método de impressão -> 0 :Transferência térmica (ribbon) 

-> 1 : Térmico Direto 
 
<ESC>PM1 
// Define o modo de impressão  -> 0 : Continuo 

     -> 1 : Rasgar      ( Tear-Off ) 
     ->  2 : Cortador      ( head position ) 
     -> 3 : Cortador      ( cutter position ) 
     -> 4 : Cortador      ( sem backfeed ) 
     -> 5 : Linerless      ( cutter position ) 
     -> 7 : Dispensador      ( head position ) 
     -> 8 : Dispensador     ( dispensing position ) 
     -> B : Cutter motion   ( cutter position + cut&print ) 
 
<ESC>PO0-00 
// Define o Offset da impressora ( POabcc ) a [Offset] 
              0 : Cortador 
              1 : Dispensador 
              2 : Rasgar ( Tear-Off ) 
              3 : Continuo 
 

b [Direcao do offset] 
              + : Feed forward ( para frente ) 
               - : Backward  ( para tras ) 
       
      cc [Valor do offset] -> ( 00 a 99 ) em dots 
 
<ESC>A3V+00000H+0000 
// Define posição de início da impressão (em dots) 
 
<ESC>CS3 
// Define a velocidade de impressão ( 2 a 10 para 203 dpi, 2 a 8 para 305 dpi, 2 a 6 para 609 dpi ) 
 
 
<ESC>#F7 
// Define a temperatura de impressão. A SATO sugere sempre permitir a variação de temperatura e 
velocidade para garantir a qualidade da impressão. 
 
<ESC>A1V00400H0799 
// Define tamanho da etiqueta (em dots na vertical e horizontal respectivamente) 
 
<ESC>Z 
// Fim de sequencia 
 
<ETX> 
// Caractere de controle que indica fim de job 

• Definição do layout da impressora e dado EPC: 
 
<STX> 

// Caractere de controle que indica início de job 
 
<ESC>A 

// Início de sequência 
 
<ESC>RU,01 

// Retorna EPC/TID gravado na tag (RU,aa) 
            aa   -> 00 : Não retorna nada (padrão) 
                  -> 01 : Retorna o dado solicitado 
 
<ESC>IP0e:h,epc:0123456789ABCDEF,fsw:0; 

// Comando que grava o dado EPC na tag em formato hexadecimal. Mais detalhes sobre 
este comando no fim deste documento. 

fsw:0 -> Permite a gravação de EPC apenas se houver campos impressos 
  fsw:1 -> Permite a gravação de EPC mesmo que não haja campos impressos 
 
<ESC>PS 

// Define caracteres com tamanhos proporcionais 
 
<ESC>%0 

// Define rotação dos objetos impressos. Neste caso, zero graus 
 
<ESC>H0155<ESC>V00078<ESC>P02<ESC>RH0,SATOSANS.ttf,0,074,079,TESTE SATO 

// Campo impresso da etiqueta 
 
 
<ESC>H0199<ESC>V00190<ESC>BG04080>I123456789012 

// Código de barras 128 impresso da etiqueta 
 
<ESC>H0301<ESC>V00274<ESC>P02<ESC>RDB@0,038,038,123456789012 

// Campo impresso da etiqueta 
 
<ESC>H0186<ESC>V00348<ESC>P02<ESC>RH0,SATOSANS.ttf,0,033,028,EPC: 

// Campo impresso da etiqueta 
 
<ESC>H0275<ESC>V00348<ESC>P02<ESC>RH0,SATOSANS.ttf,0,033,028,0123456789ABCDEF 

// Campo impresso da etiqueta 

Onde: 
 
<ESC>H (coord. Horizontal) 
<ESC>V (coord. Vertical) 
<ESC>Pxx (xx : espaçamento entre caracteres) 
<ESC>yy (yy : fonte de texto) 
   fontes proporcionais (ex : S, U, M, etc) 
   fontes vetoriais (ex : RD) complementa com [(nome da fonte), (largura), 
(altura), (texto)] 
 

Coordenadas vertical e horizontal em pontos (203 dpi = 8 pts/mm, 305dpi = 12 pts/mm) 
Espaçamento entre caracteres em pontos (idem) 
 

Continuando: 
 
<ESC>H0081<ESC>V00073<ESC>FW0303V0250H0639 

// Comando que imprime uma linha ou um retângulo na etiqueta. Neste caso foi impresso 
um retângulo. 

<ESC>FWaabccccc : Desenhar linha 
a (Largura da linha) = 02 a 99 dots 
b (Direção) 

->   H : Horizontal 
->   V : Vertical 

c (Comprimento da linha) = 01 a 832 dots 
 

<ESC>FWaabbVcccccHddddd : Desenhar um retângulo 
a (Largura Vertical da linha) = 02 a 99 dots 
b (Largura Horizontal da linha] = 02 a 99 dots 
c [Comprimento Vertical da linha] = 01 a 832 dots 
d [Comprimento Horizontal da linha] = 01 a 20000 dots 

<ESC>H0024<ESC>V00023<ESC>GB019004DADOS_IMAGEM 

//Comando que realiza a impressão de um gráfico (imagem). Além deste comando, existe 
outros 2 (dois); um que imprime imagens BMP e um que imprime imagem PCX. 
 

<ESC>Gabbbcccn...n 
a (Específica se os dados serão inseridos em formato Hexadecimal ou Binário) 

       H : Dados em Hexadecimal 
       B : Dados em binário 

b (Especificação da área gráfica transversal por byte) 
 Para range de valores, consulte manual. 
c (Especifica a área gráfica longitudinal por byte) 
 Para range de valores, consulte manual. 
n (Dados do gráfico) 

 
<ESC>Q1 

// Define quantidade de etiquetas impressas. 
 
<ESC>Z 

// Fim de sequencia 
 
<ETX> 

// Caractere de controle que indica fim de job. 
 
 

• Comando para retorno de EPC: 
 
[ DC2 ] PK 
 

// Após o envio do script de teste e, em seguida, do comando [DC2]PK, a impressora 
retorna os dados EPC no seguinte formato: 
 
[STX] a,b,c,d [CR][LF][ETX] sendo: 
 
a -> Total de bytes contidos no log (O buffer dispõe de 5KB que comportam 100 logs. Após a 
impressora retornar o log, o mesmo é apagado da memória). 
 
b -> Status de escrita 
 0 : Falha de gravação (Se não estiver setado o retorno do TID, este será o valor). 
 1 : Gravação feita com sucesso (Se estiver setado o retorno do TID, este será o valor). 

c -> Símbolo de erro 
 N : Sem erro (Será este resultado caso o retorno de TID estiver setado). 
 E : Erro ao escrever o EPC (Caso este seja o valor, a escrita de EPC não aconteceu). 

T : Erro ao ler o TID (Se estiver setado para retornar somente o EPC, esta deverá ser a 
resposta). 

 A : Erro total (Esta opção acontecera caso o log da impressora esteja vazio). 
 
d -> Retorno do dado EPC/TID 
 ee : EP (Sigla para EPC) 
  ID (Sigla para TID) 
 F : Dado EPC e(ou) TID 
 
 

//Exemplo de retorno de EPCs: 
   
165,0,T,EP:0123456789ABCDEF 
 
Onde: 
 
165 : Total de bytes contidos no log, depende da quantidade de dados EPC e(ou) TID contidos no 

log. 
 
0 : Valor de falha, mas que depende do valor de "c". Se o valor de "c" for T, então a gravação 

ocorreu com sucesso. 
 
T : Somente EPC retornado. 
 
EP : Sigla para EPC 
 
0123456789ABCDEF : EPC gravado na tag. 

• Descrição do Comando para Gravação de EPC: 
 

Situação 1 - Escrita de dados sem qualquer tipo de bloqueio na tag: 
 

<ESC>A 
<ESC>IP0e:h,epc:01234567890123456789012345670001; 
<ESC>Q1 
<ESC>Z 
 
IP0 -   Início da escrita do código EPC; 
e: -   Tipo de codificação EPC; 
h: -   Formato de codificação EPC padrão (Usado a partir do modelo CL4NX e  

    posteriores); 
,epc: -   Escrita de dado EPC; 
0123... -   Código escrito na EPC; 
; -   Fim do comando. 
 

Situação 2 - Escrita de dados com bloqueio permanente na tag: 
 
<ESC>A 
<ESC>V50<ESC>H50<ESC>XMSECURE RFID EPC DATA WRITE 
<ESC>IP0e:h,epc:01234567890123456789012345670001,lck:00001; 
<ESC>Q1 
<ESC>Z 
 
IP0 -   Início da escrita do código EPC; 
e: -   Tipo de codificação EPC; 
h: -   Formato de codificação EPC padrão (Usado a partir do modelo CL4NX e  

    posteriores); 
,epc: -   Escrita de dado EPC; 
0123... -   Código escrito na EPC. 
,lck: -   Parâmetro de bloqueio de gravação EPC; 
00001 -   Código de bloqueio de gravação EPC (Os 5 bits são obrigatórios, cada qual  

     referente a uma região de memória do chip, sendo 0 para desbloqueado, e 1  
     para bloqueado); 

; -   Fim do comando. 

Situação 3 - Escrita de dados com bloqueio de senha na tag: 
 
<ESC>A 
<ESC>V50<ESC>H50<ESC>XMSECURE RFID EPC DATA WRITE 
<ESC>IP0e:h,epc:01234567890123456789012345670001,lck:00001,apw:12345678; 
<ESC>Q1 
<ESC>Z 
 
IP0 -   Início da escrita do código EPC; 
e: -   Tipo de codificação EPC; 
h: -   Formato de codificação EPC padrão (Usado a partir do modelo CL4NX e  

     posteriores); 
,epc: -   Escrita de dado EPC; 
0123... -   Código escrito na EPC. 
,lck: -   Parâmetro de bloqueio de gravação EPC; 
00001 -   Código de bloqueio de gravação EPC (Os 5 bits são obrigatórios, cada qual  

    referente a uma região de memória do chip, sendo 0 para desbloqueado, e 1 para  
    bloqueado); 

,apw: -   Criação de senha para o bloqueio EPC; 
12345678  -   Senha de 8 dígitos escolhida (Obrigatoriamente são 8 dígitos em  

hexadecimal); 
; -   Fim do comando. 
 
 

Situação 4 - Desbloqueio e reescrita de dados em uma tag bloqueada por senha: 
 
<ESC>A 
<ESC>V50<ESC>H50<ESC>XMSECURE RFID EPC DATA WRITE 
<ESC>IP0e:h,epc:01234567890123456789012345670001,lck:00001,psw:12345678; 
<ESC>Q1 
<ESC>Z 

IP0 -   Início da escrita do código EPC; 
e: -   Tipo de codificação EPC; 
h: -   Formato de codificação EPC padrão (Usado a partir do modelo CL4NX e 
posteriores); 
,epc: -   Escrita de dado EPC; 
0123... -   Código escrito na EPC. 
,lck: -   Parâmetro de bloqueio de gravação EPC; 
00001 -   Especificando a ou as áreas bloqueadas para que o comando ,psw: saiba qual  

    área desbloquear; 
,psw: -   Desbloqueio de EPC; 
12345678 -   Senha de 8 dígitos criada com o comando ,apw:; 
; -   Fim do comando. 
 
 
 
 
 
 

Para maiores detalhes, consulte o manual de programação em SBPL. 
 


